

[Print Page](#) | [Close Window](#)

Random Usergroups error

Printed from: Snitz Forums 2000

Topic URL: http://forum.snitz.com/forum/topic.asp?TOPIC_ID=67932

Printed on: 10 February 2009

Topic:

Topic author: Webbo

Subject: Random Usergroups error

Posted on: 05 December 2008 18:04:02

Message:

Occasionally when members of a Usergroup are posting to a forum which only allows that usergroup permission to view the following error shows:

quote:

ADODB.Field error '800a0bcd'

Either BOF or EOF is True, or the current record has been deleted. Requested operation requires a current record.

/forum/inc_func_common.asp, line 1706

Line 1706 is:

```
blnAllowed = 0
Do While Not rsAllowedUserGroups.eof Or OldisAllowedMember <> 1
  Line 1706   intUserGroupID = rsAllowedUserGroups("USERGROUP_ID")
             if InStr(", " & strGroupsForUser & ", ", ", " & intUserGroupID & ",") then OldisAllowedMember = 1
             rsAllowedUserGroups.movenext
  Loop
  rsAllowedUserGroups.close
  set rsAllowedUserGroups = nothing
end if
```

The post is still made but instead of being directed to the thankyou message the poster sees the error

Any ideas?

(MySQL database)

Replies:

Reply author: ruirib

Replied on: 05 December 2008 18:53:26

Message:

```
Do While Not rsAllowedUserGroups.eof Or OldisAllowedMember <> 1
```

Not knowing the code, I would almost bet you're problem lies here, because you may have [b]OldisAllowedMember[/red] <> 1 and no records in rsAllowedUserGroups, but this is just a guess...

So is this code from the mod or was it changed by someone?

Reply author: Webbo**Replied on:** 05 December 2008 19:36:04**Message:**

That's the code from the mod

The full code for that function is:

```
function OldisAllowedMember(fForum_ID,fMemberID)
    OldisAllowedMember = 0
    if isDeniedMember(fForum_ID, fMemberID) then exit function
    strSql = "SELECT MEMBER_ID, FORUM_ID FROM " & strTablePrefix & "ALLOWED_MEMBERS "
    strSql = strSql & " WHERE FORUM_ID = " & cLng(fForum_ID)
    strSql = strSql & " AND MEMBER_ID = " & cLng(fMemberID)

    Set rsAllowedMember = Server.CreateObject("ADODB.Recordset")
    rsAllowedMember.open strSql, my_Conn
    blnAM = 0
    if not rsAllowedMember.bof and not rsAllowedMember.eof then blnAM = 1
    rsAllowedMember.close
    set rsAllowedMember = nothing

    '## if the member exists as an allowed member for the forum then return value 1
    if blnAM = 1 then
        OldisAllowedMember = 1
    else

        '## check for usergroup membership
        strGroupsForUser = getGroupMembership(fMemberID,1)
        if strGroupsForUser <> "" then
            '## check to see if the user's groups are a part of this forum
            strSql = "SELECT USERGROUP_ID FROM " & strTablePrefix & "ALLOWED_USERGROUPS "
            strSql = strSql & "WHERE FORUM_ID = " & fForum_ID
            strSql = strSql & " AND PERMS = 0"
            set rsAllowedUserGroups = my_Conn.execute(strSql)

            blnAllowed = 0
            Do While Not rsAllowedUserGroups.eof Or OldisAllowedMember <> 1
                intUserGroupID = rsAllowedUserGroups("USERGROUP_ID")
                if InStr("," & strGroupsForUser & ",", "," & intUserGroupID & ",") then OldisAllowedMember = 1
                rsAllowedUserGroups.movenext
            Loop
            rsAllowedUserGroups.close
            set rsAllowedUserGroups = nothing
        end if
    end if
end function
```

Reply author: Carefree**Replied on:** 05 December 2008 21:51:48**Message:**

On line 1705, change it to say:

Do While Not (rsAllowedUserGroups.eof or rsAllowedUserGroups.BOF) Or OldisAllowedMember <> 1

Reply author: ruirib**Replied on:** 05 December 2008 23:17:16**Message:**

quote:

Originally posted by Carefree

On line 1705, change it to say:

Do While Not (rsAllowedUserGroups.eof or rsAllowedUserGroups.BOF) Or OldisAllowedMember <> 1

This won't work, because the loop won't stop until `oldisAllowedMember` is 1.

Webbo, try this function version:

```
function OldisAllowedMember(fForum_ID,fMemberID)
    OldisAllowedMember = 0
    if isDeniedMember(fForum_ID, fMemberID) then exit function
    strSql = "SELECT MEMBER_ID, FORUM_ID FROM " & strTablePrefix & "ALLOWED_MEMBERS "
    strSql = strSql & " WHERE FORUM_ID = " & cLng(fForum_ID)
    strSql = strSql & " AND MEMBER_ID = " & cLng(fMemberID)

    Set rsAllowedMember = Server.CreateObject("ADODB.Recordset")
    rsAllowedMember.open strSql, my_Conn
    blnAM = 0
    if not rsAllowedMember.bof and not rsAllowedMember.eof then blnAM = 1
    rsAllowedMember.close
    set rsAllowedMember = nothing

    '## if the member exists as an allowed member for the forum then return value 1
    if blnAM = 1 then
        OldisAllowedMember = 1
    else

        '## check for usergroup membership
        strGroupsForUser = getGroupMembership(fMemberID,1)
        if strGroupsForUser <> "" then
            '## check to see if the user's groups are a part of this forum
            strSql = "SELECT USERGROUP_ID FROM " & strTablePrefix & "ALLOWED_USERGROUPS "
            strSql = strSql & "WHERE FORUM_ID = " & fForum_ID
            strSql = strSql & " AND PERMS = 0"
            set rsAllowedUserGroups = my_Conn.execute(strSql)

            blnAllowed = 0
            continueLoop = Not rs.AllowedUserGroups.eof

            Do While continueLoop
                intUserGroupID = rsAllowedUserGroups("USERGROUP_ID")
                if InStr(", " & strGroupsForUser & ", ", ", " & intUserGroupID & ",") then OldisAllowedMember = 1
                rsAllowedUserGroups.movenext

                continueLoop = Not rs.AllowedUserGroups.eof

                If OldisAllowedMember = 1
                    continueLoop = false
                End If
            Loop
            rsAllowedUserGroups.close
            set rsAllowedUserGroups = nothing
        end if
    end if
end function
```

Reply author: Carefree

Replied on: 06 December 2008 00:32:15

Message:

One tiny error, Ruirib, add the bit in red.

```
        If OldisAllowedMember = 1 then
            continueLoop = false
        End If
    Loop
    rsAllowedUserGroups.close
    set rsAllowedUserGroups = nothing
end if
end if
end function
```

Reply author: Webbo
Replied on: 06 December 2008 03:16:14
Message:

Hi Ruirib,

That code (with 'then' added) returns an error:

quote:

Microsoft VBScript runtime error '800a01b6'

Object doesn't support this property or method: 'rs.AllowedUserGroups'

/forum/inc_func_common.asp, line 1705

Reply author: ruirib
Replied on: 06 December 2008 07:40:14
Message:

quote:

Originally posted by Webbo

Hi Ruirib,

That code (with 'then' added) returns an error:

quote:

Microsoft VBScript runtime error '800a01b6'

Object doesn't support this property or method: 'rs.AllowedUserGroups'

/forum/inc_func_common.asp, line 1705

What line is that?

Reply author: HuwR
Replied on: 06 December 2008 07:54:16
Message:

I think the problem is that you defined the rs as
rsAllowedUserGroups but are then incorrectly referring to it as Not rs.AllowedUserGroups

Reply author: ruirib
Replied on: 06 December 2008 08:03:48
Message:

quote:

Originally posted by HuwR

I think the problem is that you defined the rs as
rsAllowedUserGroups but are then incorrectly referring to it as Not rs.AllowedUserGroups

You are right Huw, that's precisely it. Thanks.

Reply author: ruirib

Replied on: 06 December 2008 08:05:10

Message:

Webbo, here is the fixed code:

```
function OldisAllowedMember(fForum_ID,fMemberID)
    OldisAllowedMember = 0
    if isDeniedMember(fForum_ID, fMemberID) then exit function
    strSql = "SELECT MEMBER_ID, FORUM_ID FROM " & strTablePrefix & "ALLOWED_MEMBERS "
    strSql = strSql & " WHERE FORUM_ID = " & cLng(fForum_ID)
    strSql = strSql & " AND MEMBER_ID = " & cLng(fMemberID)

    Set rsAllowedMember = Server.CreateObject("ADODB.Recordset")
    rsAllowedMember.open strSql, my_Conn
    blnAM = 0
    if not rsAllowedMember.bof and not rsAllowedMember.eof then blnAM = 1
    rsAllowedMember.close
    set rsAllowedMember = nothing

    '## if the member exists as an allowed member for the forum then return value 1
    if blnAM = 1 then
        OldisAllowedMember = 1
    else

        '## check for usergroup membership
        strGroupsForUser = getGroupMembership(fMemberID,1)
        if strGroupsForUser <> "" then
            '## check to see if the user's groups are a part of this forum
            strSql = "SELECT USERGROUP_ID FROM " & strTablePrefix & "ALLOWED_USERGROUPS "
            strSql = strSql & "WHERE FORUM_ID = " & fForum_ID
            strSql = strSql & " AND PERMS = 0"
            set rsAllowedUserGroups = my_Conn.execute(strSql)

            blnAllowed = 0
            continueLoop = Not rsAllowedUserGroups.eof

            Do While continueLoop
                intUserGroupID = rsAllowedUserGroups("USERGROUP_ID")
                if InStr(", " & strGroupsForUser & ", ", ", " & intUserGroupID & ",") then OldisAllowedMember = 1
                rsAllowedUserGroups.movenext

                continueLoop = Not rsAllowedUserGroups.eof

                If OldisAllowedMember = 1 Then
                    continueLoop = false
                End If
            Loop
            rsAllowedUserGroups.close
            set rsAllowedUserGroups = nothing
        end if
    end if
end function
```

Reply author: Webbo

Replied on: 06 December 2008 12:25:11

Message:

Excellent, works a treat

Many thanks once again 😊

Edit: PS, I've just dropped a few Dollars into the kitty, it's well worth it for the support I've received 🍀

Reply author: ruirib

Replied on: 06 December 2008 15:39:54

Message:

Great 😊.

Reply author: Etymon

Replied on: 07 December 2008 01:54:52

Message:

Thanks guys for the support on this MOD! 😊

Snitz Forums 2000 : <http://forum.snitz.com/forum/>

© 2000-2009 Snitz™ Communications

[Close Window](#)